Brachistocrone problem

approximated by indirect method

```
> restart:
```

Brachistocrone functional to minimize

```
> J := sqrt( (1+y1^2)/(2*y*g)) ;
```

$$J := \frac{1}{2}\sqrt{2}\sqrt{\frac{1+y1^2}{y\,g}} \quad (1)$$

Substitution of y and y1 with related function

```
> SUBSQ := EXPR -> subs(y1=diff(y(x),x),subs(y=y(x),EXPR));
```

$$SUBSQ := EXPR \rightarrow subs\left(y1 = \frac{\mathrm{d}}{\mathrm{d}x}\,y(x),\, subs(y = y(x), EXPR)\right) \quad (2)$$

Minimize int(J(y,y'), x=0..x1)

Compute DJ/Dy1

```
> DJDY1 := diff(J,y1) ;
```

$$DJDY1 := \frac{1}{2}\,\frac{\sqrt{2}\,y1}{\sqrt{\frac{1+y1^2}{y\,g}}\;y\,g} \quad (3)$$

Compute (d/dx) (DJ/Dy1)

```
> ddx_DJDY1 := simplify(diff(SUBSQ(DJDY1),x));
```

$$ddx\_DJDY1 := -\frac{1}{4}\,\frac{\sqrt{2}\left(\left(\frac{\mathrm{d}}{\mathrm{d}x}\,y(x)\right)^2 + \left(\frac{\mathrm{d}}{\mathrm{d}x}\,y(x)\right)^4 - 2\left(\frac{\mathrm{d}^2}{\mathrm{d}x^2}\,y(x)\right)y(x)\right)}{\left(1+\left(\frac{\mathrm{d}}{\mathrm{d}x}\,y(x)\right)^2\right)y(x)^2\,g\,\sqrt{\frac{1+\left(\frac{\mathrm{d}}{\mathrm{d}x}\,y(x)\right)^2}{y(x)\,g}}} \quad (4)$$

Compute DJ/Dy

```
> DJDY := SUBSQ(diff(J,y)) ;
```

$$DJDY := -\frac{1}{4}\,\frac{\sqrt{2}\left(1+\left(\frac{\mathrm{d}}{\mathrm{d}x}\,y(x)\right)^2\right)}{\sqrt{\frac{1+\left(\frac{\mathrm{d}}{\mathrm{d}x}\,y(x)\right)^2}{y(x)\,g}}\;y(x)^2\,g} \quad (5)$$

Eulero lagrange function

```
> ODE := simplify(DJDY - ddx_DJDY1);
```

(6)

$$ODE := -\frac{1}{4}\,\frac{\sqrt{2}\left(1+\left(\frac{d}{dx}y(x)\right)^2+2\left(\frac{d^2}{dx^2}y(x)\right)y(x)\right)}{\left(1+\left(\frac{d}{dx}y(x)\right)^2\right)y(x)^2 g\sqrt{\frac{1+\left(\frac{d}{dx}y(x)\right)^2}{y(x)\,g}}} \tag{6}$$

Simplified ODE

```
> newODE := -numer(ODE)/sqrt(2) ;
```

$$newODE := 1+\left(\frac{d}{dx}y(x)\right)^2+2\left(\frac{d^2}{dx^2}y(x)\right)y(x) \tag{7}$$

Finite difference approximation of the ODE

```
> FD := k -> 1+((Y[k+1]-Y[k-1])/(2*h))^2 + 2*Y[k]*(Y[k+1]+Y[k-1]
  -2*Y[k])/h^2 ;
```

$$FD := k \to 1+\frac{1}{4}\,\frac{\left(Y_{k+1}-Y_{k-1}\right)^2}{h^2}+\frac{2\,Y_k\left(Y_{k+1}+Y_{k-1}-2\,Y_k\right)}{h^2} \tag{8}$$

N internal discretization

```
> N := 40 ;
  h := 3/N ;
```

$$N := 40$$

$$h := \frac{3}{40} \tag{9}$$

The nonlinear system

```
> EQ0  := Y[0]:
  for k from 1 to 100 do
    EQ||k := FD(k):
  end:
  EQ||N := Y[N]+1:
```

Solution of the nonlinear system

```
> RES := fsolve( {EQ||(0..N)},{seq(Y[k]=-k*h,k=0..N)} ):
```

Plot of the solution

```
> plot( subs(RES,[seq([k*h,Y[k]],k=0..N)]), scaling=CONSTRAINED )
  ;
```


0
1
2
3
−0,2
−0,4
−0,6
−0,8
−1,0
−1,2